


Worst Case Analysis for an Analog Filter

In this application, the worst case analysis for Analog bandpass filter is performed. The circuit is an active filter with Op-amp. And, the list of data calculated from the nominal value and the tolerance for constants of electrical components is used for the analysis.



1. Functions to Calculate Transfer function

Impedance of Capacitor
in Laplace domain

$$Z_c := (C_0) \rightarrow \frac{1}{s \cdot C_0}$$

Impedance of Inductor
in Laplace domain

$$Z_L := (L_0) \rightarrow s \cdot L_0$$

Impedance combination
of Parallel components

$$11_z := (Z_1, Z_2) \rightarrow \frac{Z_1 \cdot Z_2}{Z_1 + Z_2}$$

2. Transfer function of Analog bandpass filter

Input impedance $Z_i := R_1 + Z_c(C_1) = R_1 + \frac{1}{s \cdot C_1}$

Feedback impedance $Z_f := 11_z(R_2, Z_c(C_2)) = \frac{R_2}{s \cdot C_2 \cdot \left(R_2 + \frac{1}{s \cdot C_2} \right)}$

Transfer function $G_s := -\frac{Z_f}{Z_i} = -\frac{R_2}{s \cdot C_2 \cdot \left(R_2 + \frac{1}{s \cdot C_2} \right) \cdot \left(R_1 + \frac{1}{s \cdot C_1} \right)}$

Create System object for Maple's DynamicSystems package $\text{tfsys} := \text{DynamicSystems}:-\text{TransferFunction}(G_s)$

Check the object

$\text{DynamicSystems}:-\text{PrintSystem}(\text{tfsys}) =$

Transfer Function
 continuous
 1 output(s); 1 input(s)
 inputvariable = [u1(s)]
 outputvariable = [y1(s)]

$$\text{tf}_{1,1} = -\frac{R_2 \cdot C_1 \cdot s}{C_1 \cdot C_2 \cdot R_1 \cdot R_2 \cdot s^2 + (C_1 \cdot R_1 + C_2 \cdot R_2) \cdot s + 1}$$

3. Create the parameter list

Parts name $\text{Parts} := [R_1, R_2, C_1, C_2]$

Nominal value $\text{Nom} := [4.7 \cdot 10^3, 47 \cdot 10^3, 10 \cdot 10^{-9}, 680 \cdot 10^{-12}]$

Tolerance $\text{Tol} := \left[10, 5, 5, \frac{1}{680} \cdot 100 \right]$

List of parameters based on tolerance

$\text{Par} := \left[\text{seq}(\text{Nom} + \sim \text{Nom} \cdot \sim \text{Tol} \cdot \sim (\text{Bits}:-\text{Split}(i, \text{bits} = \text{nops}(\text{Nom})) \cdot 2 - \sim 1) \cdot 0.01, i = 1 \dots \text{nops}(\text{Nom})^2) \right]$

4. Bode Plot


Obtain the plot data of magnitude & phase plots

```
maglist := [seq(DynamicSystems:-BodePlot(tfsys,  
  parameters = [seq(Parts[i] = Par[j][i], i = 1 ..nops(Parts))],  
  output = magnitudeplot, hertz), j = 1 ..nops(Nom)^2)]
```


```
phaseplist := [seq(DynamicSystems:-BodePlot(tfsys,  
  parameters = [seq(Parts[i] = Par[j][i], i = 1 ..nops(Parts))],  
  output = phaseplot, hertz), j = 1 ..nops(Nom)^2)]
```

Check plots

```
plots:-display(maglist) =
```



```
plots:-display(phaseplist) =
```



5. Find the set of parameters for Max and Min

Obtain the list of data of magnitude plot

```
magdat := [seq(DynamicSystems:-BodePlot(tfsys,  
  parameters = [seq(Parts[i] = Par[j][i], i = 1 ..nops(Parts))],  
  output = magnitudedata, hertz), j = 1 ..nops(Nom)^2)]
```

Collect the datapoints of the maximum magnitude for each parameter set

```
maxmag := [seq(max(convert(magdat[i][1..-1, 2], list)), i = 1 ..nops(Nom)^2)]
```

Find the parameter sets which have the maximum and minimum point

```
maxvalue, maxpos := ListTools:-FindMaximalElement(maxmag, position)
```

```
minvalue, minpos := ListTools:-FindMinimalElement(maxmag, position)
```

Organize results

```
MaxPattern := seq(Parts[i] = Par[maxpos][i], i = 1 ..nops(Parts))
```

```
NomPattern := seq(Parts[i] = Nom[i], i = 1 ..nops(Parts))
```

```
MinPattern := seq(Parts[i] = Par[minpos][i], i = 1 ..nops(Parts))
```

Final result of Maximum/Nominal/Minimum

MaxPattern = $R_1 = 4.230 \times 10^3$, $R_2 = 4.935 \times 10^4$, $C_1 = 1.050 \times 10^{-8}$, $C_2 = 6.790 \times 10^{-10}$

NomPattern = $R_1 = 4.700 \times 10^3$, $R_2 = 47000$, $C_1 = 1.000 \times 10^{-8}$, $C_2 = 6.800 \times 10^{-10}$

MinPattern = $R_1 = 5.170 \times 10^3$, $R_2 = 4.465 \times 10^4$, $C_1 = 9.500 \times 10^{-9}$, $C_2 = 6.810 \times 10^{-10}$